

**PRACTICE EXERCISES OF THE MICROPROCESSORS
& MICROCONTROLLERS**

Instructor: The Tung Than

Student's name: Gia Hung Nguyen


Student code: 24520603

**PRACTICE REPORT NO 3
USING INTERRUPT**

- I. Lưu đồ thuật toán xử lý nút nhấn**
- II. Thiết kế phần cứng**
- III. Xây dựng chương trình Assembly**
- IV. Kết quả mô phỏng**

I. Lưu đồ thuật toán xử lý nút nhấn

1. Lưu đồ thuật toán



Hình 1. Lưu đồ giải thuật hồi vòng tuần tự 4 nút nhấn A, B, C, D

2. Giải thích

Chương trình của đồng hồ thể thao chạy như sau:

- **Ngắt Timer 0:** Đảm nhận duy nhất việc đếm thời gian. Cứ mỗi 10ms, Timer sẽ tự động ngắt và cập nhật các thanh ghi chứa dữ liệu thời gian.
- **Vòng lặp chính:** Thực hiện liên tục việc xuất dữ liệu ra 4 LED 7 đoạn (Quét LED) và kiểm tra trạng thái 4 nút nhấn (A, B, C, D) thông qua cơ chế Polling.

Chức năng 4 nút:


- **Nút A (Pause/Resume):** Sử dụng cơ chế đảo trạng thái của bit điều khiển Timer (TR0). Bằng lệnh CPL TR0, nếu TR0 = 1 (đang chạy), nhấn nút sẽ xóa TR0 = 0 (tạm dừng). Ngược lại, nếu TR0 = 0, nhấn nút sẽ đặt TR0 = 1 (tiếp tục đếm).
- **Nút B (Reset):** Khi được nhấn, chương trình sẽ ép dừng Timer (CLR TR0) và xóa các thanh ghi lưu trữ giá trị thời gian về 0 để đưa đồng hồ về mốc 00.00.

- **Nút C (Tăng 1s):** Tăng giá trị thanh ghi lưu phần Giây lên 1 đơn vị bằng lệnh INC R1. Nếu giá trị đang ở mức 99, chương trình đưa đồng hồ về mức 00.00.
- **Nút D (Giảm 1s):** Giảm giá trị thanh ghi lưu phần Giây đi 1 đơn vị bằng lệnh DEC R1. Nếu giá trị đang ở mức 00, chương trình đưa đồng hồ về mức 00.00.

II. Thiết kế phần cứng

Mô phỏng được thực hiện trên phần mềm Proteus với các thành phần chính:

- **Khối điều khiển trung tâm:** Vi điều khiển họ 8051 (IC AT89C51).
- **Đồng hồ thể thao:** Gồm 4 LED 7 đoạn dùng để hiển thị thời gian từ 00.00 đến 99.99 giây. Dữ liệu BCD xuất qua IC giải mã 74247 nối với **Port 2**. Các chân cấp nguồn quét LED nối với **Port 1**.
- **Khối Nút nhấn:** Gồm 4 nút nhấn điều khiển nối vào **Port 3**:
 - Nút A: Tạm dừng/Tiếp tục đếm.
 - Nút B: Reset về 00.00.
 - Nút C: Tăng 1 giây.
 - Nút D: Giảm 1 giây.



Hình 2. Sơ đồ mạch đồng hồ thể thao với 4 LED 7 đoạn trên Proteus

III. Xây dựng chương trình Assembly

ASSEMBLY	GIẢI THÍCH
<pre>ORG 0000H LJMP START ORG 000BH LJMP ISR_TIMER0 START: CLR P1.4 CLR P1.5 MOV TMOD, #11H MOV TH0, #0D8H MOV TLO, #0F0H SETB ET0 SETB EA CLR TR0 MOV R1, #0 MOV R2, #0</pre>	Khởi tạo hệ thống & Timer: Xóa chân P1.4 và P1.5 để chuẩn bị đo Oscilloscope. Cấu hình Timer: TMOD = 11H (Cả Timer 0 và 1 đều hoạt động ở Chế độ 1 - 16-bit). Nạp giá trị đếm D8F0H cho Timer 0 tương đương thời gian trễ đúng 10ms. Bật cờ ngắt ET0 và EA. Trạng thái ban đầu: Tắt Timer 0 (CLR TR0) chờ ấn nút A. R1 (Giây) = 0, R2 (Phần trăm giây) = 0.

ASSEMBLY	GIẢI THÍCH
<pre> MAIN_LOOP: ACALL HIEN_THI ACALL CHECK_BUTTON SJMP MAIN_LOOP ISR_TIMER0: CLR TR0 MOV TH0, #0D8H MOV TL0, #0F0H CPL P1.4 SETB TR0 INC R2 CJNE R2, #100, EXIT_ISR MOV R2, #0 INC R1 CJNE R1, #100, EXIT_ISR MOV R1, #0 EXIT_ISR: RETI </pre>	Vòng lặp chính: Liên tục quét LED và kiểm tra trạng thái nút nhấn. Ngắt Timer 0 (Đếm thời gian 10ms): Mỗi khi tràn, ngắt tạm dừng để nạp lại giá trị D8F0H. Lệnh CPL P1.4 đảo trạng thái chân tạo ra xung vuông đo trên Oscilloscope. Sau đó, tăng R2 lên 1. Nếu đủ 100 (1 giây) thì reset R2 = 0 và cộng dồn lên R1. Nếu R1 đủ 100 giây thì quay về 00.
<pre> CHECK_BUTTON: JNB P3.2, BUTTON_A JNB P3.3, BUTTON_B JNB P3.4, BUTTON_C JNB P3.5, BUTTON_D RET BUTTON_A: JNB P3.2, \$ CPL TR0 RET BUTTON_B: JNB P3.3, \$ CLR TR0 MOV R1, #0 MOV R2, #0 RET </pre>	Quản lý Nút bấm (Polling): Lệnh JNB P3.x, \$ chờ người dùng nhấn phím ra thì chương trình mới thực thi lệnh (chống dội phím). Nút A (P3.2 - Pause/Resume): Dùng lệnh CPL TR0 để đảo trạng thái chạy/dừng của Timer đếm giờ. Nút B (P3.3 - Reset): Ép Timer dừng đếm (CLR TR0) và xóa sạch các biến thời gian R1, R2 về 0.

ASSEMBLY	GIẢI THÍCH
BUTTON_C: <pre> JNB P3.4, \$ INC R1 CJNE R1, #100, EXIT_BUTTON SJMP RESET_VE_0000 </pre> BUTTON_D: <pre> JNB P3.5, \$ MOV A, R1 JZ RESET_VE_0000 DEC R1 RET </pre> RESET_VE_0000: <pre> MOV R1, #0 MOV R2, #0 </pre> EXIT_BUTTON: <pre> RET </pre>	Nút C (P3.4 - Tăng 1s): Tăng biến R1 lên 1 (INC R1). Nếu đang ở 99 mà ấn tiếp thì sẽ nhảy thẳng đến RESET_VE_0000 để đưa về 00.00. Nút D (P3.5 - Giảm 1s): Giảm biến R1 đi 1 (DEC R1). Đặc biệt áp dụng chốt chặn: Lệnh JZ kiểm tra nếu đang ở 00 mà ấn tiếp thì sẽ nhảy thẳng đến RESET_VE_0000 để giữ ở mốc 00.00.
HIEN_THI: <pre> MOV A, R1 MOV B, #10 DIV AB MOV P2, A SETB P2.4 MOV P1, #00000001b ACALL DELAY_1MS MOV P1, #0 MOV P2, B CLR P2.4 MOV P1, #00000010b ACALL DELAY_1MS MOV P1, #0 </pre>	Tách & Quét LED Giây (R1): Dùng lệnh DIV AB (chia cho 10) để tách hàng chục và đơn vị. Xuất Hàng chục ra P2. Dùng SETB P2.4 để tắt dấu chấm phân cách. Bật LED qua P1.0. Xuất Hàng đơn vị ra P2. Dùng CLR P2.4 để bật sáng điểm chấm thập phân, tạo ra hiển thị dạng XX.XX. Sau mỗi nhịp 1ms, luôn tắt P1 = 0 để chống hiện tượng bóng ma (ghosting).

ASSEMBLY	GIẢI THÍCH
<pre> MOV A, R2 MOV B, #10 DIV AB MOV P2, A SETB P2.4 MOV P1, #00000100b ACALL DELAY_1MS MOV P1, #0 MOV P2, B SETB P2.4 MOV P1, #00001000b ACALL DELAY_1MS MOV P1, #0 RET </pre>	Tách & Quét LED Trăm giây (R2): Lặp lại thao tác chia DIV AB cho biến phần trăm giây R2. Tại 2 vị trí LED này, hoàn toàn không cần dấu chấm hiển thị phân cách nên chương trình liên tục gọi lệnh SETB P2.4 để dập tắt DP. Mỗi khi thực hiện xong quét 4 LED, lệnh RET sẽ kết thúc hàm, trả CPU quay lại MAIN_LOOP để quét phím.
<pre> DELAY_1MS: CPL P1.5 MOV TH1, #0FCH MOV TL1, #018H CLR TF1 SETB TR1 JNB TF1, \$ CLR TR1 RET END </pre>	Hàm Delay 1ms bằng Timer 1: Lệnh CPL P1.5 đảo trạng thái chân tạo xung để theo dõi trên Oscilloscope. Nạp giá trị đếm FC18H (64536) vào Timer 1. Thời gian đếm cần để tràn cờ là: $65536 - 64536 = 1000$ chu kỳ máy (tương đương chính xác 1ms). Lệnh JNB TF1, \$ giữ CPU chờ (Polling) cho đến khi cờ tràn TF1 bật lên. Sau đó tắt Timer 1 và thoát hàm, giúp độ trễ nháy LED.

IV. Kết quả mô phỏng

Link video mô phỏng nguyên lý hoạt động của mạch và File thiết kế Proteus:

<https://drive.google.com/drive/folders/13KdWxi85grbi3sJfzdUt4mdF8npo3Wh1?usp=sharing>